

# Budget Challenge Solution

---

*By Neale Blackwood*

*[www.a4accounting.com.au](http://www.a4accounting.com.au)*

*[a4@iinet.net.au](mailto:a4@iinet.net.au)*



**A4 ACCOUNTING**  
EXCEL CONSULTING, TRAINING AND WEBINARS

## Contents

Introduction .....	3
Data Sheet.....	4
Interface Sheet.....	5
Template .....	5
Template Creation .....	6
Workings .....	6
Template Copying .....	12
Interface Total Company Report.....	13
Workings .....	13
Total Sector Reports .....	14
Workings .....	14
3-D formulas .....	14
Validations .....	15
Validate total unit sales .....	15
Validate sector unit sales .....	16
Overall Validation.....	16
Conditional format.....	16
Other sheets.....	16

## Introduction

In early November 2018 I became aware of a challenge to create a budget based on a scenario set by an academic who wanted to research various solutions to budgeting problems in Excel.

I submitted a solution in the middle of November and was granted permission from the academic, Paul Mireault, to create a free webinar based on my solution.

I am not suggesting that this is the only solution or the best solution, but it is a working solution to the problem.

The solution combines a number of Excel techniques that can be applied to other budgeting and forecasting solutions and I thought given the existence of the detailed scenario that this could become a useful webinar.

I assume you have already read through the scenario (provided separately) and I will only be discussing a few problematic issues from the scenario that need to be addressed.

I have provided the Excel solution file that I submitted as well as the blank version. You can use the blank file to create your own solution. This session will focus on the solution file and explain its structure and formulas.

The challenge was basically a four-dimension profit and loss budget where we needed to separate sales into sectors, products, regions and months.

The Excel file supplied contained two sheets

- **Data** – contains all the specifications mentioned in the scenario description
- **Interface** - the structure of the report that was required

Since the Interface sheet had the desired reporting structure I decided to use that as my template to build the budget.

The interface sheet already had three of the four dimensions contained on it

1. Products
2. Regions
3. Months

Because of this I decided to use Excel's 3-D formula functionality and create a separate sheet for each sector/region combination, which was the fourth dimension that was not contained on the Interface sheet.

Excel has the ability to perform calculations through multiple sheets and we'll see this demonstrated in the webinar.

**WARNING:** this is not a beginner's session and includes some advanced topics. I won't be going into detail about every function used in this solution.

	A	B	C	D	E	F	G	H	I	J
1	Interface						Jan	Feb	Mar	Apr
2										
3	Enter Base Price:	\$100.00		Product	Regions	N				
4				S		SE				
5	Total Profit					SW				
6						E				
7						W				
8										
9				Product	Regions	N				
10				D		SE				
11						SW				
12						E				
13						W				
14										
15				Unit Sales by	Product	S				
16				Month and Product	D					
17										
18				Unit Sales by Month						
19										
20				Sales Amount by	Product	S				
21				Month and Product	D					
22										
23				Profit by Month						
24										

## Data Sheet

The Data sheet contains all of the inputs that we require to create the budget with the exception of the sales price which is on the Interface sheet.

The data sheet contains multiple tables. Tables are the best way to handle inputs because many of Excel's functions work well with tables.

The shaded cells were input cells.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Data												
2													
3	Monthly Fixed Cost	\$20,000											
4													
5	Product	Standard	Deluxe										
6	Base Price Multiplier	1	1.45										
7	Unit Production Cost	\$48.00	\$72.00										
8													
9	Product Distribution per Sector												
10		Standard	Deluxe	Total									
11	Government	65%	35%	100%									
12	Military	25%	75%	100%									
13	Private	40%	60%	100%									
14	Education	80%	20%	100%									
15													
16													
17													
18	Region	North	South-Eas	South-West	East	West							
19	Unit Delivery Cost	\$10.25	\$9.73	\$9.58	\$8.26	\$11.02							
20													
21													
22													
23													
24													
25													
26													
27													
28	Sector	Rebate											
29	Government	Percentage	DemParA	DemParB									
30	Military	40%	3.59	22000000000									
31	Private	20%	3.46	22000000000									
32	Education	10%	3.18	22000000000									
		70%	4.11	22000000000									

	Sector	Government	Military	Private	Education
Region	N	25%	52%	22%	24%
	SE	18%	13%	21%	15%
	SW	18%	18%	17%	32%
	E	22%	0%	25%	17%
	W	17%	17%	15%	12%
	Total	100%	100%	100%	100%

	Sector	Government	Military	Private	Education
Month	Jan	9%	8%	12%	6%
	Feb	10%	9%	11%	8%
	Mar	12%	10%	9%	9%
	Apr	12%	12%	7%	10%
	May	11%	13%	6%	12%
	Jun	9%	11%	4%	12%
	Jul	7%	9%	5%	11%
	Aug	6%	7%	6%	9%
	Sep	5%	6%	8%	7%
	Oct	5%	4%	9%	6%
	Nov	6%	5%	11%	5%
	Dec	8%	6%	12%	5%
	Total	100%	100%	100%	100%

### Issues

The one thing I didn't like about the structure of the Data sheet was the fact that the data tables are under one another.

In general I find it is much easier to work with tables going across the page with a blank column between each table. This structure makes handling and expanding the tables much easier.

Given this was a challenge and the file was going to be compared to other solutions I didn't re-organise the table.

**Note:** cell K8 has zero, this means there are no sales for that sector/region combination.

## Interface Sheet

As mentioned the Interface sheet contains three of the four dimensions that we need to handle. I will use this sheet as the template to create individual sheets for each sector/region and the total sheets.

The four dimension we need to handle are

Product – Widget	Regions	Sectors	Months
Standard (S)	N	Government	Jan
Deluxe (D)	SE	Military	Feb
	SW	Private	Mar
	E	Education	Apr
	W		May
			Jun
			Jul
			Aug
			Sep
			Oct
			Nov
			Dec

## Template

Because the sales for a sector are affected by the region I decided to create a separate sheet for each combination of sector and region. I also decided to use a code in the sheet name to simplify the creation of the other sector/region sheets.

I decided to use a prefix and suffix to identify the sector and the region respectively. As an example the Government - North combination will have a sheet name of G\_N.

In Excel you can create a formula to extract the sheet name. My solution will use the sheet name to automatically create the sector/region report once the sheet is renamed.

The template sheet will have the report structure in the top section and the bottom section will contain all the workings required to calculate the sales based on the sheet name which determines the sector and the region.

This structure allows me to create one sector/region template sheet and then copy and rename it multiple times to create all of the other combinations for sector and region.

There will also be a number of other blank sheets created which will be placeholder (boundary) sheets. These sheets will be explained later in the session as we discuss the 3-D formula functionality of Excel.

## Template Creation

The Interface sheet will be copied and used as a template. The original Interface sheet will also be used to summarise all of the other template sheets that will be created. It will be renamed Interface Total Company. This sheet will have the input of the base price in cell B3.

Each sector/region report template sheet will be identical but changing the name of the sheet will automatically create the sector/region report. I will use sheet colour coding to identify each region. Each sector will have five separate sheets to handle all five regions.

The top section (rows 1 to 23) of each template sheet will have the existing report structure. This will contain formulas that will extract the relevant information from the Workings section below.

In the completed file I have included an Instructions sheet which includes this explanation of the sheet naming convention that I used.

The completed file also has an Index sheet to assist with sheet navigation.

Sector and Region sheets (no inputs required)	
Individual sheets create the Sector by Region calculations	
The are are two abbreviations used in the sheet names	
They are separated by the underscore character	
The start of the sheet name has the Sector letter	
G = Government	Green Sheets
M = Military	Blue sheets
P = Private	Orange sheet
E = Education	Yellow sheets
The end of the sheet name has the region	
N = North	
SE = South East	
SW = South West	
E - East	
W = West	

Cell B3 in the sector/region Template sheet will be linked to cell B3 in the Interface Total Company sheet.

## Workings

Row 30 downwards will include the Workings section for each sector/region combination and will extract all of the necessary values from the Data sheet and perform all the necessary calculations.

Column A contain labels and column B contains formulas.

Let's work our way through the Workings section.

### Formula - cell B31 (Path & Filename)

=CELL("filename",A1)

This extracts the full pathname including the current sheet name on the end.

The important part of this formula is including a cell from this sheet as the second argument. This ensures the cell will be accurate. Omitting the second argument can display incorrect results.

We need to extract the sheet name from the end of cell B31.

	A	B
30	<b>Workings</b>	<b>Calculated</b>
31	Path & Filename	E:\A4\Training\W
32	Sheet Name	G_N
33	Sector Code	G
34	Region Code	N
35	Region Name	North
36	Sector	Government
37	Rebate %	40%
38	Sales Price	60.00
39	DemParA	3.59
40	DemParB	22000000000
41	Total Demand	9096.193533
42	Sector %	25%
43	Standard	65%
44	Deluxe	35%

**Formula – cell B32 (Sheet Name)**

=RIGHT(B31,LEN(B31)-SEARCH("]",B31))

This formula looks for the ending square bracket in cell B31 and it extracts everything from the right of the square bracket which is the sheet name.

The RIGHT function extracts characters from the right of a cell. The LEN function counts how many characters are in a cell. The SEARCH function returns a number that represents the character position of what is being searched for. The difference between the results of the LEN and SEARCH functions will be the length of the sheet name.

Now that we have extracted the sheet name we need to extract the prefix and the suffix to determine the sector and the region.

**Formula – cell B33 (Sector Code)**

=LEFT(B32,SEARCH("\_",B32)-1)

This formula extracts the characters from the left of the underscore character in cell B32. The LEFT function extracts from the left of the cell. The reason we subtract one from the SEARCH result as we do not require the underscore character.

**Formula – cell B34 (Region Code)**

=RIGHT(B32,LEN(B32)-SEARCH("\_",B32))

This is the same type of formula used in cell B32 to extract the sheet name from the right of the string in cell B31. The formula in B34 extracts the characters from the right of the underscore character in cell B32.

Having extracted the region code we will convert it into the region name.

**Data Sheet**

To enable us to use these sector and region codes I added some extra codes to the Data sheet. They are highlighted in yellow in the image on the right.

	A	B	C	D	E	F
10	Sector	Standard	Deluxe	Total	Sector Code	
11	Government	65%	35%	100%	G	
12	Military	25%	75%	100%	M	
13	Private	40%	60%	100%	P	
14	Education	80%	20%	100%	E	
15						
16						
17	Region Code	N	SE	SW	E	W
18	Region	North	South-Eas	South-West	East	West
19	Unit Delivery Cost	\$10.25	\$9.73	\$9.58	\$8.26	\$11.02
20						

**Formula – cell B35 (Region Name)**

=INDEX(Data!\$B\$18:\$F\$18,MATCH(B34,Data!\$B\$17:\$F\$17,0))

The INDEX function allows you to extract data from a range based on a row number and a column number. In the above case we are extracting the full region name from row 18 of the Data sheet based on the code in row 17 in the Data sheet.

If the range in an INDEX function is a single row or a single column you only need to provide a single number to extract from that range. In our case we are extracting from a single row, row 17 in the Data sheet, so we only need to provide the column number to extract from that range. The MATCH function returns a number that represents the position of what is being matched in the range that it is being matched in.

#### **Formula – cell B36 (Sector)**

=INDEX(Data!\$A\$11:\$A\$14,MATCH(B33,Data!\$E\$11:\$E\$14,0))

This formula is extracting the full name of the sector in column A based on the code in column E both in the Data sheet. See the image on the previous page.

Now that we have identified which sector we are working in we can use the rest of Row 36 in the workings section to return the monthly sales split percentage. The month columns in the top section starts in column G. The workings section at the bottom also uses the same month columns.

#### **Formula – cell G36 (Monthly Sales % Split - Jan)**

=INDEX(Data!\$J\$15:\$M\$26,MATCH(G\$1,Data!\$I\$15:\$I\$26,0),  
MATCH(\$B\$36,Data!\$J\$14:\$M\$14,0))

This formula extracts the monthly sales percentage for the sector based on the table on the Data sheet. The first MATCH function provides the row number to extract. The second MATCH function provides the column number to extract. The INDEX function takes the intersection of the row number and the column number to extract from the two-dimensional range.

This formula can be copied across the remainder of rows 36 until column R.

#### **Formula – cell B37 (Rebate %)**

=INDEX(Data!\$B\$29:\$B\$32,MATCH(\$B\$36,Data!\$A\$29:\$A\$32,0))

Each sector has its own rebate. This formula extracts the rebate to use for the sector which is based on a table in the Data sheet. Each sector has its own rebate. The rebate is a discount percentage based on the full price that is an input in the Interface Total Company sheet.

#### **Formula – cell B38 (Sales Price)**

=B3\*(1-B37)

This formula calculates the sector's sales price by adjusting the base sales price by the sector's rebate percentage. **Note** cell B3 is linked to cell B3 in the Interface Total Company sheet. That is where the base sales price is entered.

#### **Demand Calculation**

A formula was provided to calculate the demand in units for the products. The formula has two inputs that relate to the demand as well as the sales price as an input. The two demand inputs are

provided in a table on the Data sheet. We will extract both of them and perform the calculation for the demand in units.

**Formula – cell B39 (DemParA)**

=INDEX(Data!\$C\$29:\$C\$32,MATCH(\$B\$36,Data!\$A\$29:\$A\$32,0))

**Formula – cell B40 (DemParB)**

=INDEX(Data!\$D\$29:\$D\$32,MATCH(\$B\$36,Data!\$A\$29:\$A\$32,0))

Now that we have the three components to calculate demand we can do the demand calculation in cell B41.

**Note:** This is the total demand in units for the full year for the whole sector.

**Formula – cell B41 (Total Demand)**

=B40/B38^B39

The ^ symbol means to the power of (indices) and is calculated first. So the first calculation that is performed is the selling price to the power of 3.59 (Demand Part A). Then Demand Part B is divided by that calculated number to arrive at the total demand in units based on price and sector.

We need to allocate the total demand in units for the sector but to do that we need to determine the sector's percentage.

**Formula – cell G41 (Monthly Total Units)**

=+\$B\$41\*G36

This allocates total sales units across all the months using the percentage split in row 36.

**Formula – cell B42 (Total Demand)**

=INDEX(Data!\$J\$5:\$M\$9,MATCH(B34,Data!\$I\$5:\$I\$9,0),  
MATCH(B36,Data!\$J\$4:\$M\$4,0))

This formula extracts the sector/region percentage from the table on the Data sheet – see image on the right.

This percentage represents the percentage of total unit sales for the sector that are made in the region.

	H	I	J	K	L	M
3	Region Sales Distribution per Sector					
4		Sector	Government	Military	Private	Education
5	Region	N	25%	52%	22%	24%
6		SE	18%	13%	21%	15%
7		SW	18%	18%	17%	32%
8		E	22%	0%	25%	17%
9		W	17%	17%	15%	12%
10		Total	100%	100%	100%	100%
11						

As an example 25% of the total Government sales are made in the North region.

**Formula – cell G42 (Monthly Total Region Units)**

=+\$B\$42\*G41

This takes the total sector unit sales in row 41 and allocates the region's monthly unit sales based on the percentage in cell B42.

**Formula – cell B43 (Standard)**

=INDEX(Data!\$B\$11:\$B\$14,MATCH(\$B\$36,Data!\$A\$29:\$A\$32,0))

**Formula – cell B44 (Deluxe)**

=INDEX(Data!\$C\$11:\$C\$14,MATCH(\$B\$36,Data!\$A\$29:\$A\$32,0))

These formulas extract the percentage sales for the standard and deluxe products for the sector.

**Formula – cell G43 & G44 (Standard + Deluxe Units)**

=+\$B43\*G\$42

This formula has been copied down and across and calculates the split between standard and deluxe units.

**Formula – cell G45 (Check Units)**

=G42-SUM(G43:G44)

This formula validates the split of the sales between standard and deluxe and makes sure it equals the total from row 42 (Monthly Total Region Units). It should display a zero. This row will be used in by the Validations sheet.

**Unit Selling Prices cell B47 & B48 (Standard and Deluxe Sales \$)**

The unit prices are determined by direct links to the base price in cell D3 or the deluxe factor from cell C6 (value 1.45) in the Data sheet.

**Formula – cell G47 & G48 (Standard and Deluxe Sales \$)**

=+G43\*\$B47\*(1-\$B\$37)

This formula calculates the monthly sales and has been copied down and across. It calculates the standard and deluxe sales dollars based on the unit sales, the unit selling prices and the rebate for the sector.

Rows 51 to 54 calculate the units costs and have straight forward formulas – there is a separate cost for Standard and Deluxe – both costs come from the Data sheet.

Row 56 calculates delivery costs. The Data sheet has an input cell for a \$ rate per unit sold.

Row 57 calculates the Total Variable costs by adding the product's unit cost and the delivery cost.

Rows 58 to 60 allocate Fixed Costs (\$20,000 per month). Based on sales units. I chose to allocate fixed costs based on total sales units as there were no rules or instructions provided with this scenario. This allocation needs all sector/region sheets to be created before it will work as it needs

to calculate total sales. This formula will be explained later as we go through the webinar. You will understand it after we go through the total sheets.

Row 62 calculates the total profit.

## The Report

Now we have all the figures we require to populate the report at the top of the sheet.

### **Formula – cell G3 (Product Standard unit sales in North region)**

**=(\$F3=\$B\$34)\*G\$43**

This formula will include the unit sales from row 43 if the sheet is for the North region. All other regions will display zeroes. This formula can be copied across and down for Standard units.

The logical test in the bracket compares cell F3 to B43. If they are the same it will return TRUE.

In Excel TRUE = 1 and FALSE =0.

If you multiply by TRUE you are multiplying by one and the number remains the same.

If you multiply by FALSE you are multiplying by zero and you zero the value.

You must enclose the logical test in brackets to use this technique. When zeroing values the above structure is effective.

You could have used the following IF function to achieve the same result.

**=IF(\$F3=\$B\$34,G\$43,0)**

### **Formula – cell G9 (Product Deluxe unit sales in North region)**

**=(\$F9=\$B\$34)\*G\$44**

This is the same structure as the formula in cell G3, but it returns the Deluxe units. This formula can be copied across and down for Deluxe units.

Rows 15, 16 and 18 summarise the unit sales above.

Rows 20, 21 and 23 link directly to the rows in the Workings section for Sales by product and total Profit.

Cell B5 adds up all the monthly profits from row 23.

## Template Copying

Now that we have one complete Template, we need to create 19 other templates to handle the other combinations of Sector and Region.

I have decided to keep the sheets grouped by Sector rather than Region.

Below is the final layout with all the sheets created. Sorry it is difficult to read



G = Government
M = Military
P = Private
E = Education

The black sheets on the far left and far right are a and z respectively.

They are used to add up the total company values. You will see the formulas that use them soon.

The grey sheets, for example Ga and Gz, allow us to add up just the Government sheets.

Each sector has their own pair of grey sheet surrounding them.

The other grey sheets allow us to add up the other sectors.

All the black and grey sheets are empty.

Let's see how we use them.

## Interface Total Company Report

The original Interface sheet was used to create the Sector/Region Template. It is also the basis of the Interface Total Company sheet.

This Interface Total Company sheet must be positioned on the left of sheet a or to right of sheet z.

The report section has a single formula that can be copied across and down to populate the whole report. We will paste just the formula rather than the formats.


### Formula – cell G3 (Product Standard unit sales Total company)

=SUM(a:z!G3)

You could type this formula or use the mouse. I will demonstrate both techniques in the webinar.

This adds up cells G3 in every sheet from a to z including a and z.

The a and z sheets define the boundaries of the sheets to include in the formula.



You can copy the above formula and use Paste Formulas (see image on right) to paste in all the other cell requiring values.

## Workings

As well as the summary report at the top of the sheet I have also used the same formula and copied it to the Workings section of the sheet. These rows line up with the rows in the sector/region sheets.

This provides extra detail for the total sales and costs.

You can see the breakdown of the individual rows in the Workings section.

## Total Sector Reports

I decided to include total reports for each sector and used the original interface sheet as the basis for these reports.

Whereas the Interface Total Company sheet formulas used the black sheets a and z, the sector reports use the grey sheets to identify the boundaries of the sheets to add up.

As an example the formula below relates to the Interface Total Government sheet.

### **Formula – cell G3 (Product Standard unit sales Total Government)**

**=SUM(Ga:Gz!G3)**

This adds up all the G3 cells in all the sheets between Ga and Gz (including Ga and Gz) which includes all of the government sector sheets. The Ga and Gz sheets are both blank. These are boundary sheets for the government reports.

Again, this formula can be copied across and down to the other value rows.

You just need to make sure you use paste formulas instead of the normal paste.

### **Workings**

As well as the summary report at the top of the sheet I have also used the same formula and copied it to the Workings section of the sheet. These rows line up with the rows in the sector/region sheets.

This provides extra detail for the sector sales and costs.

You can see the breakdown of the individual rows in the Workings section.

### **3-D formulas**

These formulas use the 3-D structure that allows you to add up multiple sheets.

**WARNING:** Using these types of formula is the only time that the location of a sheet is important. You shouldn't put any other sheets between a and z. These sheets act as the boundaries of the budget sheets.

## Validations

As I was building the model I identified a number of areas that required validating. These related to the Data sheet as well as the totals for each of the sectors.

The validations sheet (image below) has a number of validations listed as well as an overall validation.

I recommend including an overall validation that you can link to your report sheets so that you know that the model is validating and report is accurate.

	A	B	C	D	E	F	G	H
1	<b>Validations</b>							
2	<b>Overall Validation</b>	OK						
3			1	2	3	4		
4	Product Distribution per Sector = 100%	TRUE	TRUE	TRUE	TRUE	TRUE		
5	Region Sales Distribution per Sector = 100%	TRUE	TRUE	TRUE	TRUE	TRUE		
6	Monthly Sales Distribution per Sector = 100%	TRUE	TRUE	TRUE	TRUE	TRUE		
7	Units Allocated - Total = 0	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE
8	Units Allocated - Government = 0	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE
9	Units Allocated - Military = 0	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE
10	Units Allocated - Private = 0	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE
11	Units Allocated - Education = 0	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE
12								

When validating I display TRUE if the cell is valid and FALSE if it isn't.

Column B contains summary validations that I will explain later.

### Formula – cell C4 (Product Distribution per Sector)

=IFERROR(INDEX(Data!\$D\$11:\$D\$14,C3)=1,"")

This formula extracts the product distribution per sector and makes sure that it equals 100%. If it does it will display TRUE if it doesn't it displays FALSE. If there is an error it will display a blank cell. I have used sequential helper cells in row 3 to simplify extracting the values using the INDEX function.

Rows 5 and 6 have similar formulas. They confirm that Region Sales are allocated 100% (row 5) and Monthly sales are allocated 100% (row 6).

### Validate total unit sales

Row 7 validates the units allocated in total. This is based on row 45 in the Interface Total Company sheet.

=Interface Total Company!G45=0

This is in the Workings section of the sheet and confirms the individual product allocations (standard and deluxe) equal the total units sold. This will display TRUE if it equals zero and FALSE if it doesn't.

## Validate sector unit sales

Rows 8 to 11 use the same type of formula but they check each sector's total sheet.

Column B has formulas that check all of the other validation cells.

### Formula – cell B3 (Product Distribution per Sector)

**=COUNTIF(C4:N4,FALSE)=0**

This formula counts the number of times FALSE appears in the range C4 to N4. If it equals zero then it will display TRUE. Otherwise it will display FALSE. This formula has been copied down the column.

This allows you to check many cells for FALSE.

If the report have many columns the validation cells on a row may not display on a single screen and hence may not be visible. This formula summarises them all in one cell.

## Overall Validation

Cell B2 has an overall validation that displays OK or Error. The cell is linked to the other reporting sheets and summarises all the other validations.


### Formula – cell B2 (Overall Validation)

**=IF(COUNTIF(B4:B11,FALSE)=0,"OK","Error")**

This uses the same COUNTIF function but checks column B. If no FALSE entries are found then OK is displayed. Otherwise Error is displayed.

## Conditional format

I have used the following conditional format on cell B2.



## Other sheets

I included an Index sheet to make navigation around the model easier and I also included an Instructions sheet with explanations.